clear all
set more off
capture log close
pause on


/* SC internal notes*/

* This .do file starts with the csv file that has articles about the ICC and Catholic Church.  This .csv is first created with catholicchurchr3_2021_09_17.R.  I then manually coded the 237 articles.  Coding rules are currently described in the RR draft memo.
*	It is only drug articles that have a church search term (catholic bishop archbishop) or an ICC term (bensouda ICC international criminal court).

* cd "/Users/robertchaudoin/Dropbox/Kill_Scrapes/Media coverage data/"
* cd "/Users/stephenchaudoin/Dropbox/Kill_Scrapes/Media coverage data/"
* Commands that deleted the columns that can't be shared on Dataverse
* import delimited using "d.drugbefore_churchicc_2021_09_17", clear
* drop data
* drop source_url
* export delimited using "churchicc_replication", replace

import delimited using "churchicc_replication", clear

* Tidying
replace contestation = "ICC only" if contestation == "iCC only"
drop if contestation == "ICC only" | contestation == "Not relevant"
destring contestation, replace

gen epubdate = mdy(month, day, year)
gen epubdate2 = epubdate
format epubdate %dM_d,_CY	

gen one = 1
bysort epubdate: egen churcharticlesday_ = total(one)
bysort epubdate: egen churchconarticlesday_ = total(contestation)

duplicates drop epubdate, force

keep epubdate church*_* year month day

tsset epubdate

* This is the list of church "events"
* 8/16/2017: Kian dls killed
* 9/11/2017: Catholic Bishops Council of the Philippines (CBCP) calls for a stop to the killings.
*	epubdate == 21073
* 9/14/2017: Caloocan Bishop speaks out.  Church bell ritual.
* 21076
* 9/18/2017: Cubao Bishop starts scholarship for kids of WOD victims.
* 21080
* 11/5/2017: CBCP has 33 day "healing protest" regarding the WOD
* 21128
* 11/08/2017: CBCP criticizes the firing of a Dangerous Drug Board official who was thought to be less rabid about the WOD.
* 21131
* 12/06/2017: Cebu Arch Bishop Palma makes a relatively tame statement about cops respecting human life.
* 21159
* 12/08/2017: Diocese of Legazpi forms a campaign against the WOD.
* 21161
* 12/10/2017: Church leader expresses hesitation about transfer of WOD back to PNP, away from PDEA.
* 21163
* 12/12/2017: Church leader speaks out against bounties for killing suspects.
* 21165
* 1/26/2018: Davao Archbishop calls for cops to respect due process.
* 21210
* 1/30/2018: Church in Bacolod calls for a bloodless resumption of Oplan Tokhang.
* 21214
* 2/24/2018: CBCP organizes a Walk for Life against the killings.
* 21239


twoway (bar churcharticlesday_ epubdate, bcolor(gs10)) (bar churchconarticlesday_ epubdate, bcolor(gs2)), ///
	ytitle("Number of Articles") xtitle("")  xlab(,labsize(vsmall) angle(-90)) ///
	xline(21073 21076 21080 21128 21131 21159 21161 21163 21165 21210 21214 21239, lwidth(0.2)) ///
	ttext(7 21073 "CBCP Speaks Out, Church Bell Ritual" 5 21128 "33 Days of Healing", size(vsmall)) ///
	ttext(7 21165 "Church leaders speak out against resuming Oplan Tokhang", size(vsmall)) ///
	ttext(5 21214 "Various Church leaders speak out", size(vsmall)) ///
	ttext(4 21239 "CBCP organizes Walk for Life", size(vsmall)) ///
	graphregion(color(white)) bgcolor(white) ///
	legend(r(2) label(1 "Articles w/ Church content") label(2 "Articles w/ Church content + Contestation"))

	
	
